@ECHO OFF
SETLOCAL

:: This command launches a Visual Studio solution with environment variables required to use a local version of the .NET SDK.

:: This tells .NET to use the same dotnet.exe that build scripts use
SET DOTNET_ROOT=%~dp0.dotnet
SET DOTNET_ROOT(x86)=%~dp0.dotnet\x86

:: Put our local dotnet.exe on PATH first so Visual Studio knows which one to use
SET PATH=%DOTNET_ROOT%;%PATH%

SET sln=%~1

IF "%sln%"=="" (
    echo Error^: Expected argument ^<SLN_FILE^>
    echo Usage^: startvs.cmd ^<SLN_FILE^>

    exit /b 1
)

IF NOT EXIST "%DOTNET_ROOT%\dotnet.exe" (
    echo .NET has not yet been installed. Run `%~dp0restore.cmd` to install tools
    exit /b 1
)

IF "%VSINSTALLDIR%" == "" (
    start "" "%sln%"
) else (
    "%VSINSTALLDIR%\Common7\IDE\devenv.com" "%sln%"
)
